

9/11.2.3

Second jeu de fonctions évoluées

Dans ce paragraphe, nous vous proposons quatre fonctions destinées à améliorer le fonctionnement du traitement de texte WEKA étudié à la Partie 9, Chap. 11.2. En voici la liste :

- déplacement rapide du curseur à la page précédente ;
- déplacement rapide du curseur à la page suivante ;
- effacement du mot à droite du curseur ;
- effacement du mot à gauche du curseur.


Ces quatre fonctions sont exécutées par des sous-programmes écrits en Assembleur. Elles se logent immédiatement après la fin du premier jeu de fonctions évoluées : adresse &H8363.

Examinons en détails les diverses actions effectuées par chacune des nouvelles fonctions et donnons pour chacune l'analyse du sous-programme correspondant, extrait du programme général.

Déplacement rapide du curseur à la page précédente

Lorsque l'utilisateur appuie simultanément sur les touches **Ctrl** et flèche vers le haut, la page précédente est affichée (si elle existe). Le curseur est affiché à la même position.

Cette action obéit à la logique suivante :




Le sous-programme correspondant à cette fonction a pour nom TPGUP.

- Lignes 35 et 36 : Sauvegarde adresse curseur.
- Lignes 39 à 48 : Calcul adresse début écran.
- Lignes 50 à 56 : 24 lignes plus haut.
- Ligne 57 : Adresse convenable.
- Lignes 59 à 86 : Affichage de la page précédente.
- Lignes 87 et 88 : Restitution du curseur.
- Lignes 89 à 99 : Restitution adresse ligne courante.

Déplacement rapide du curseur à la page suivante

Lorsque l'utilisateur appuie simultanément sur les touches **Ctrl** et flèche vers le bas, la page suivante est affichée (si elle existe). Le curseur est affiché à la même position.

Cette action obéit à la logique suivante :




Le sous-programme correspondant à cette fonction a pour nom TPGDN.

- Lignes 107 et 108 : Sauvegarde adresse du curseur.
- Lignes 111 à 122 : Calcul adresse début écran.
- Lignes 127 à 130 : 24 lignes plus bas.
- Lignes 131 à 152 : Affichage de la page suivante.
- Lignes 153 et 154 : Restitution du curseur.
- Lignes 155 à 165 : Restitution adresse ligne courante.

Effacement du mot à droite du curseur

Lorsque l'utilisateur appuie simultanément sur les touches **Ctrl** et **T**, le mot (ou la portion de mot) qui se trouve à droite du curseur est effacé. Le curseur conserve sa position. Aucun mot n'est effacé dans le cas où le curseur se trouve en fin de ligne.

Cette action obéit à la logique suivante :




Le sous-programme correspondant à cette fonction a pour nom TEFMD.

- Lignes 172 et 173 : Sauvegarde coordonnées curseur.
- Lignes 174 et 175 : Curseur en ligne 78.
- Lignes 178 à 217 : Recherche du début du mot à droite.
- Lignes 218 à 236 : Effacement d'un caractère.
- Ligne 237 : Affichage de la nouvelle ligne.

Effacement du mot à gauche du curseur

Lorsque l'utilisateur appuie simultanément sur les touches **Ctrl** et **R**, le mot (ou la portion de mot) qui se trouve à gauche du curseur est effacé. Le curseur conserve sa position. Aucun mot n'est effacé dans le cas où le curseur se trouve en début de ligne.

Cette action obéit à la logique suivante :



Le sous-programme correspondant à cette fonction a pour nom TEFMG.

- Lignes 248 et 249 : Sauvegarde coordonnées curseur.
- Lignes 250 et 251 : Curseur en ligne 1.
- Lignes 254 à 316 : Recherche du début du mot à gauche.
- Lignes 319 à 335 : Effacement d'un caractère.
- Ligne 336 : Affichage de la nouvelle ligne.

Le programme Assembleur qui contient donc ces quatre fonctions a été écrit sous ZEN. En voici le listing :

```

1          ORG  8363H
2          LOAD 8363H
3          ;-----
4          ;Suite 2 Traitement de texte WEKA
5          ;-----
6          ; Fonctions rajoutees:
7          ; Ctrl Up : Une page plus haut
8          ; Ctrl Down : Une page plus bas
9          ; Ctrl T : Effacement mot a droite
10         ; Ctrl R : Effacement mot a gauche
11         ;
12         ;=====
13         ; SUITE BOUCLE PRINCIPALE
14         ;=====
15         ;
16 8363 FEF8          CP   CTRLUP
17 8365 2812         JR   Z,TPGUP          ;Une page plus haut
18 8367 FEF9          CP   CTRLDN
19 8369 CAF283       JP   Z,TPGDN          ;Une page plus bas
20 836C FE14         CP   CTRLT
21 836E CA5D84       JP   Z,TEFMD          ;Effect mot droite
22 8371 FE12         CP   CTRLR
23 8373 CAD584       JP   Z,TEFMD          ;Effect mot gauche
24 8376 C32F82       JP   AFFICH          ;Affichage caractere
25         ;-----
26         ;
27         ;-----
28         ; ZONE DES TRAITEMENTS
29         ;-----

```

```

30      ;
31      ;-----
32      ;Deplact une page plus haut
33      ;-----
34      TPGUP:      EQU  $
35 8379 CD78BB      CALL GETCUR      ;Position curseur
36 837C 228782      LD  (SAVCUR),HL      ;Sauv @ curseur
37 837F 45          LD  B,L          ;Ligne courante
38 8380 05          DEC  B
39 8381 2A8082      LD  HL,(ADRLIG)      ;@ debut ligne
40 8384 78          LD  A,B
41 8385 B7          OR  A
42 8386 2809        JR  Z,B0SUI      ;Curseur sur 1ere ligne
43 8388 115000      LD  DE,80          ;80 caract par ligne
44 838B 37          SCF
45 838C 3F          CCF          ;Plus de retenue
46          B0:      EQU  $
47 838D ED52        SBC  HL,DE
48 838F 10FC        DJNZ B0          ;Recherche @ deb ecran
49          B0SUI:   EQU  $
50 8391 118007      LD  DE,1920        ;24 lignes + haut
51 8394 37          SCF
52 8395 3F          CCF
53 8396 ED52        SBC  HL,DE
54 8398 228582      LD  (SAUV2),HL
55 839B 110090      LD  DE,DEBTEX
56 839E ED52        SBC  HL,DE
57 83A0 384D        JR  C,NOAC      ;Pas d'action
58
59 83A2 3E02        LD  A,2
60 83A4 CD0EBC      CALL MODE      ;Effact ecran
    
```

```

61          ;
62 83A7 2A8582      LD   HL,(SAUV2)
63          ;
64 83AA 228082      LD   (ADRLIG),HL      ;Sauvegarde
65 83AD 0619        LD   B,25          ;Nb lignes a afficher
66 83AF 210101      LD   HL,101H
67 83B2 228382      LD   (SAUVHL),HL      ;Coord curseur
68          BOUAF0: EQU  $          ;Boucle affichage
69 83B5 C5          PUSH BC          ;Sauvegarde B
70 83B6 E5          PUSH HL          ; Sauveg HL
71 83B7 0650        LD   B,80         ;Nb car par ligne
72 83B9 2A8082      LD   HL,(ADRLIG)
73          BOUW0:  EQU  $
74 83BC 7E          LD   A,(HL)
75 83BD CD5ABB      CALL WRCHAR
76 83C0 23          INC  HL
77 83C1 10F9        DJNZ BOUW0
78 83C3 228082      LD   (ADRLIG),HL      ;Sauv @ deb ligne
79 83C6 2A8382      LD   HL,(SAUVHL)
80 83C9 2C          INC  L
81 83CA 2601        LD   H,1
82 83CC 228382      LD   (SAUVHL),HL
83 83CF CD75BB      CALL SETCUR
84 83D2 E1          POP  HL
85 83D3 C1          POP  BC
86 83D4 10DF        DJNZ BOUAF0          ;Boucle d'affichage
87 83D6 2A8782      LD   HL,(SAVCUR)
88 83D9 CD75BB      CALL SETCUR          ;Restitution curseur
89 83DC 3E19        LD   A,25
90 83DE 95          SUB  L
91 83DF 47          LD   B,A

```

```

92 83E0 115000      LD  DE,80
93 83E3 2A80B2      LD  HL,(ADRLIG)
94 83E6 3F          SCF
95 83E7 3F          CCF
96                  B20:  EQU  $
97 83E8 ED52        SBC  HL,DE
98 83EA 10FC        DJNZ B20
99 83EC 2280B2      LD  (ADRLIG),HL
100                 NOAC:  EQU  $
101 83EF C31A80      JP  ACQCAR          ;Fin du traitement
102                 ;
103                 ;-----
104                 ;Deplact une page plus bas
105                 ;-----
106                 TPGDN:  EQU  $
107 83F2 CD78BB      CALL GETCUR          ;Position curseur
108 83F5 228782      LD  (SAVCUR),HL     ;Sauv @ curseur
109 83F8 45          LD  B,L             ;Ligne courante
110 83F9 05          DEC  B
111 83FA 2A80B2      LD  HL,(ADRLIG)     ;@ debut ligne
112 83FD 78          LD  A,B
113 83FE B7          OR  A
114 83FF 2809        JR  Z,B1SUI          ;Curseur sur 1ere ligne
115 8401 115000      LD  DE,80           ;80 caract par ligne
116 8404 3F          SCF
117 8405 3F          CCF                ;Plus de reter:ue
118                 B1:    EQU  $
119 8406 ED52        SBC  HL,DE
120 8408 10FC        DJNZ B1             ;Recherche @ deb ecran
121                 B1SUI: EQU  $

```

```

122 840A E5          PUSH HL          ;Sauveg HL
123                ;
124 840B 3E02        LD    A,2
125 840D CD0EBC      CALL MODE      ;Effact ecran
126                ;
127 8410 E1          POP  HL          ;Restit HL
128 8411 118007      LD    DE,1920  ;24 lignes + bas
129 8414 19          ADD  HL,DE
130 8415 228082      LD    (ADRLIG),HL ;Sauvegarde
131 8418 0619        LD    B,25     ;Nb lignes a afficher
132 841A 210101      LD    HL,101H
133 841D 228382      LD    (SAUVHL),HL ;Coord curseur
134                BOUAF: EQU  $          ;Boucle affichage
135 8420 C5          PUSH BC          ;Sauvegarde B
136 8421 E5          PUSH HL          ; Sauveg HL
137 8422 0650        LD    B,80     ;Nb car par ligne
138 8424 2A8082      LD    HL,(ADRLIG)
139                BOUW: EQU  $
140 8427 7E          LD    A,(HL)
141 8428 CD5ABB      CALL WRCHAR
142 842B 23          INC  HL
143 842C 10F9        DJNZ BOUW
144 842E 228082      LD    (ADRLIG),HL ;Sauv @ deb ligne
145 8431 2A8382      LD    HL,(SAUVHL)
146 8434 2C          INC  L
147 8435 2601        LD    H,1
148 8437 228382      LD    (SAUVHL),HL
149 843A CD75BB      CALL SETCUR
150 843D E1          POP  HL
151 843E C1          POP  BC
152 843F 10DF        DJNZ BOUAF      ;Boucle d'affichage

```

```

153 8441 2A8782      LD   HL, (SAVCUR)
154 8444 CD75BB      CALL SETCUR          ;Restitution curseur
155 8447 3E19        LD   A,25
156 8449 95          SUB  L
157 844A 47          LD   B,A
158 844B 115000      LD   DE,80
159 844E 2A8082      LD   HL, (ADRLIG)
160 8451 37          SCF
161 8452 3F          CCF
162                B2:   EQU  $
163 8453 ED52        SBC  HL,DE
164 8455 10FC        DJNZ B2
165 8457 228082      LD   (ADRLIG),HL
166 845A C31A80      JP   ACQCAR          ;Fin du traitement
167                ;
168                ;-----
169                ;Effacement du mot a droite
170                ;-----
171                TEFMD: EQU  $
172 845D CD78BB      CALL GETCUR
173 8460 22B382      LD   (SAUVHL),HL    ;@ curseur
174 8463 7C          LD   A,H
175 8464 FE4E        CP   78
176 8466 D2D284      JP   NC,FINMD       ;Effect impossible
177                ; Recherche debut mot a droite
178 8469 CD7082      CALL RESTIT         ;Restit car courant
179 846C CD78BB      CALL GETCUR         ;Position curseur
180 846F 25          DEC  H
181 8470 CD75BB      CALL SETCUR         ;Restit pos curseur
182                DROI0: EQU  $

```

```
183 8473 CD60BB      CALL RDCHAR          ;Lecture car courant
184 8476 FE20        CP   SPACE           ;Blanc?
185 8478 2015        JR   NZ,DROI1        ;Non
186                  DROI2: EQU  #
187 847A CD78BB      CALL GETCUR
188 847D 7C          LD   A,H
189 847E FE4F        CP   79
190 8480 281E        JR   Z,DROI3        ;Fin de deplact
191 8482 24          INC  H
192 8483 CD75BB      CALL SETCUR
193 8486 CD60BB      CALL RDCHAR
194 8489 FE20        CP   SPACE
195 848B 28ED        JR   Z,DROI2        ;Boucle
196 848D 1811        JR   DROI3         ;Fin de deplact
197
198                  DROI1: EQU  #
199 848F CD78BB      CALL GETCUR
200 8492 7C          LD   A,H
201 8493 FE4F        CP   79             ;Fin de ligne ?
202 8495 2809        JR   Z,DROI3        ;Oui
203 8497 CD78BB      CALL GETCUR
204 849A 24          INC  H
205 849B CD75BB      CALL SETCUR
206 849E 18D3        JR   DROI0         ;Boucle
207                  DROI3: EQU  #
208 84A0 CD78BB      CALL GETCUR
209 84A3 ED5B6382    LD   DE,(SAUVHL)
210 84A7 7C          LD   A,H
211 84A8 92          SUB  D
212 84A9 2827        JR   Z,FINMD        ;Pas possible
213 84AB F5          PUSH AF
```

```

214 84AC 2A83B2          LD   HL, (SAUVHL)
215 84AF CD75BB          CALL SETCUR
216 84B2 F1              POP  AF
217 84B3 47              LD   B,A
218                      BOUMD: EQU  $
219 84B4 C5              PUSH BC
220                      ;
221 84B5 CD78BB          CALL GETCUR
222 84B8 3E50            LD   A,80
223 84BA 94              SUB  H
224 84BB 47              LD   B,A
225 84BC 04              INC  B
226 84BD 1600            LD   D,0
227 84BF 5C              LD   E,H
228 84C0 2A80B2          LD   HL, (ADRLIG)
229 84C3 19              ADD  HL,DE
230 84C4 2B              DEC  HL
231 84C5 54              LD   D,H
232 84C6 5D              LD   E,L
233 84C7 23              INC  HL
234                      TCLR1: EQU  $
235 84C8 EDA0            LDI
236 84CA 10FC            DJNZ TCLR1
237 84CC CD52B2          CALL WRILIG           ;Aff nle ligne
238                      ;
239 84CF C1              POP  BC
240 84D0 10E2            DJNZ BOUMD           ;Boucle d'effacement
241                      FINMD: EQU  $
242 84D2 C31A80          JP   ACQCAR           ;Fin du traitement
243                      ;

```

```

244      ;-----
245      ;Effacement du mot a gauche
246      ;-----
247      TEFMG:      EQU  $
248 84D5 CD78BB      CALL GETCUR
249 84D8 228382      LD   (SAUVHL),HL      ;@ curseur
250 84DB 7C          LD   A,H
251 84DC FE01        CP   1
252 84DE CA7485      JP   Z,FINMG      ;Effect impossible
253      ;Recherche debut mot a gauche
254 84E1 CD7082      CALL RESTIT      ;Caract courant
255 84E4 CD78BB      CALL GETCUR
256 84E7 25          DEC  H
257 84E8 CD75BB      CALL SETCUR      ;Restit pos courante
258      GAUS0:      EQU  $
259 84EB CD60BB      CALL RDCHAR      ;Lect car courant
260 84EE FE20        CP   SPACE      ;Blanc ?
261 84F0 281C        JR   Z,GAU2      ;Oui
262 84F2 CD78BB      CALL GETCUR
263 84F5 7C          LD   A,H
264 84F6 FE01        CP   1
265 84F8 284E        JR   Z,GAU4      ;Plus de deplact
266 84FA 25          DEC  H
267 84FB CD75BB      CALL SETCUR
268 84FE CD60BB      CALL RDCHAR
269 8501 FE20        CP   SPACE
270 8503 2802        JR   Z,GAU0
271 8505 1815        JR   GAU3
272      GAU0:      EQU  $
273 8507 CD60BB      CALL RDCHAR      ;Lect car courant
274 850A FE20        CP   SPACE      ;Blanc ?

```

```

275 850C 200E      JR   NZ,GAU3      ; Non
276                GAU2: EQU  #
277 850E CD78BB    CALL GETCUR
278 8511 7C        LD   A,H
279 8512 FE01      CP   1
280 8514 2832      JR   Z,GAU4      ;Plus de deplact
281 8516 25        DEC  H
282 8517 CD75BB    CALL SETCUR
283 851A 18EB      JR   GAU0        ;Boucle
284                GAU3: EQU  #
285 851C CD78BB    CALL GETCUR
286 851F 7C        LD   A,H
287 8520 FE01      CP   1
288 8522 2824      JR   Z,GAU4      ;Plus de deplact
289 8524 25        DEC  H
290 8525 CD75BB    CALL SETCUR
291 8528 CD60BB    CALL RDCHAR
292 852B FE20      CP   SPACE      ;Blanc ?
293 852D 20ED      JR   NZ,GAU3      ;Non
294 852F CD78BB    CALL GETCUR
295 8532 24        INC  H
296 8533 CD75BB    CALL SETCUR      ;Ajustement curseur
297 8536 1810      JR   GAU4        ;Fin du traitement
298                GAU1: EQU  #
299 8538 CD60BB    CALL RDCHAR
300 853B FE01      CP   1
301 853D 2809      JR   Z,GAU4      ;Plus de deplact
302 853F CD78BB    CALL GETCUR
303 8542 25        DEC  H
304 8543 CD75BB    CALL SETCUR

```

```

305 8546 18BF          JR   GAU0           ;Boucle de recherche
306                   GAU4:   EQU  $           ;Fin de traitement
307                   ;
308 8548 CD78BB        CALL GETCUR
309 854B 7C            LD   A,H           ;Pos courante
310 854C 2A8382        LD   HL,(SAUVHL)
311 854F 6F            LD   L,A
312 8550 7C            LD   A,H
313 8551 95            SUB  L
314 8552 47            LD   B,A           ;Nbre de car a effacer
315 8553 B7            OR   A
316 8554 281E         JR   Z,FINMG        ;Rien a effacer
317                   ;
318                   BOUMG:  EQU  $           ;Boucle effect
319 8556 C5            PUSH BC
320 8557 CD78BB        CALL GETCUR
321 855A 3E50          LD   A,80
322 855C 94            SUB  H
323 855D 47            LD   B,A
324 855E 04            INC  B
325 855F 1600         LD   D,0
326 8561 5C            LD   E,H
327 8562 2A8082        LD   HL,(ADRLIG)
328 8565 19            ADD  HL,DE
329 8566 2B            DEC  HL
330 8567 54            LD   D,H
331 8568 5D            LD   E,L
332 8569 23            INC  HL
333                   TCLR2:  EQU  $
334 856A EDA0          LDI

```

```

335 856C 10FC          DJNZ TCLR2
336 856E CD5282       CALL WRILIG          ;Aff nlle ligne
337 8571 C1          POP BC
338 8572 10E2       DJNZ BOUMG          ;Tant que B<>0
339                FINMG:    EQU  #
340 8574 C31A80       JP ACQCAR          ;Fin du traitement
341                ;
342                ;=====
343                ;      ZONE DES EQU
344                ;=====
345                ;
346                CTRLUP:    EQU 248
347                CTRLDN:    EQU 249
348                CTRLT:     EQU 20
349                CTRLR:     EQU 18
350                ;
351                AFFICH:    EQU 822FH          ;Ref 1er prog
352                ACQCAR:    EQU 801AH          ;Ref 1er prog
353                RESTIT:    EQU 8270H          ;Ref 1er prog
354                ADRLIG:    EQU 8280H          ;Ref 1er prog
355                WRILIG:    EQU 8252H          ;Ref 1er prog
356                GETCUR:    EQU 0BB78H          ;TXT GET CURSOR
357                SETCUR:    EQU 0BB75H          ;TXT SET CURSOR
358                WRCHAR:    EQU 0BB5AH          ;TXT OUTPUT
359                MODE:      EQU 0BC0EH          ;SCR SET MODE
360                SAUVHL:    EQU 8283H
361                SAUV2:     EQU 8285H          ;Sauvegarde HL
362                SAVCUR:    EQU 8287H          ;Sauv @ curseur
363                DEBTEX:    EQU 9000H          ;Debut du texte
364                RDCHAR:    EQU 0BB60H          ;TXT RD CHAR
365                SPACE:     EQU 32
366                END

```

Comme toujours, voici la version Chargeur Basic correspondant au programme Assembleur listé ci-dessus. Cette version inclut également le programme de base et le premier jeu de fonctions.

```

1000 REM =====
1010 REM Traitement de texte avec complements 1 et 2
1020 REM =====
1030 FOR I=&8000 TO &827F
1040   READ A$
1050   A$="&"+A$
1060   POKE I,VAL(A$)
1070 NEXT I
1080 '
1090 REM -----
1100 REM Donnees du programme principal
1110 REM -----
1120 DATA 3E,2,CD,E,BC,21,0,90,22,80,82,21,0,90,1,0
1130 DATA 10,16,20,72,B,23,73,B1,20,F9,CD,8A,BB,CD,6,BB
1140 DATA FE,D,CA,55,80,FE,F0,CA,8A,80,FE,F1,CA,DE,80,FE
1150 DATA F2,CA,11,81,FE,F3,CA,28,81,FE,1A,CA,9D,81,FE,17
1160 DATA CA,37,81,FE,10,CA,E8,81,FE,7F,CA,5,82,FE,11,C8
1170 DATA C3,2F,82,18,C5,CD,70,82,2A,80,82,11,50,0,19,22
1180 DATA 80,82,CD,78,BB,7D,FE,19,28,C,CD,78,BB,26,1,2C
1190 DATA CD,75,BB,C3,1A,80,6,1,AF,CD,4D,BC,CD,52,82,CD
1200 DATA 78,BB,26,1,CD,75,BB,C3,1A,80,CD,70,82,CD,78,BB
1210 DATA 7D,FE,1,28,16,2A,80,82,11,50,0,ED,52,22,80,82
1220 DATA CD,78,BB,2D,25,CD,75,BB,C3,1A,80,21,0,90,ED,5B
1230 DATA 80,82,ED,52,20,A,CD,78,BB,25,CD,75,BB,C3,1A,80
1240 DATA 6,0,AF,CD,4D,BC,2A,80,82,11,50,0,ED,52,22,80
1250 DATA 82,CD,52,82,CD,78,BB,25,CD,75,BB,C3,1A,80,CD,70
1260 DATA 82,2A,80,82,11,50,0,19,22,80,82,CD,78,BB,7D,FE
1270 DATA 19,28,B,CD,78,BB,25,2C,CD,75,BB,C3,1A,80,6,1
1280 DATA AF,CD,4D,BC,CD,52,82,CD,78,BB,25,CD,75,BB,C3,1A
1290 DATA 80,CD,78,BB,7C,FE,1,CA,1A,80,CD,70,82,CD,78,BB
1300 DATA 25,25,CD,75,BB,C3,1A,80,CD,78,BB,7C,FE,50,CA,1A
1310 DATA 80,CD,70,82,C3,1A,80,CD,78,BB,22,87,82,45,2A,80
1320 DATA 82,11,50,0,37,3F,ED,52,10,FC,19,22,83,82,11,0
1330 DATA 90,ED,52,7C,B5,CA,1A,80,CD,78,BB,6,0,4C,2A,80
1340 DATA 82,9,2B,7E,CD,5A,BB,6,0,AF,CD,4D,BC,21,1,1
1350 DATA CD,75,BB,2A,80,82,22,85,82,2A,83,82,11,50,0,ED
1360 DATA 52,22,80,82,CD,52,82,2A,87,82,CD,75,BB,2A,85,82
1370 DATA 11,50,0,37,3F,ED,52,22,80,82,C3,1A,80,CD,78,BB
1380 DATA 22,87,82,6,0,4C,2A,80,82,9,2B,7E,CD,5A,BB,6
1390 DATA 1,AF,CD,4D,BC,21,19,1,CD,75,BB,2A,80,82,22,85
1400 DATA 82,ED,5B,87,82,3E,1A,93,47,11,50,0,19,10,FD,22
1410 DATA 80,82,CD,52,82,2A,85,82,11,50,0,19,22,80,82,2A
1420 DATA 87,82,CD,75,BB,C3,1A,80,CD,78,BB,3E,50,94,47,4
1430 DATA 16,0,5C,2A,80,82,19,2B,54,5D,23,ED,A0,10,FC,CD
1440 DATA 52,82,C3,1A,80,CD,78,BB,7C,FE,1,CA,1A,80,3E,50
1450 DATA 94,47,4,16,0,5C,2A,80,82,19,2B,54,5D,1B,ED,A0

```

```

1460 DATA 10,FC,CD,52,82,CD,78,BB,25,CD,75,BB,C3,1A,80,FE
1470 DATA 20,DA,1A,80,32,82,82,CD,78,BB,4C,7C,FE,50,CA,1A
1480 DATA 80,6,0,2A,80,82,9,2B,3A,82,82,77,CD,5A,BB,C3
1490 DATA 1A,80,CD,78,BB,22,83,82,26,1,CD,75,BB,6,50,2A
1500 DATA 80,82,7E,CD,5A,BB,23,10,F9,2A,83,82,CD,75,BB,C9
1510 DATA CD,78,BB,4C,6,0,2A,80,82,9,2B,7E,CD,5A,BB,C9
1520 '
1530 '-----
1540 ' Modification du 1er programme
1550 '-----
1560 '
1570 POKE &8051,&89 : POKE &8052,&82
1580 '
1590 '=====
1600 ' Suite 1 de l'editeur de texte WEKA
1610 '=====
1620 FOR I=&8289 TO &8362
1630   READ A#
1640   A=VAL("&" + A#)
1650   POKE I,A
1660 NEXT I
1670 '
1680 '-----
1690 ' Donnees de la suite 1
1700 '-----
1710 DATA FE,1,CA,A3,82,FE,6,CA,B,83,FE,FB,CA,45,83,FE
1720 DATA FA,CA,54,83,C3,2F,82,C3,1A,80,CD,70,82,CD,78,BB
1730 DATA 25,CD,75,BB,CD,60,BB,FE,20,28,1C,CD,78,BB,7C,FE
1740 DATA 1,28,3C,25,CD,75,BB,CD,60,BB,FE,20,28,2,18,15
1750 DATA CD,60,BB,FE,20,20,E,CD,78,BB,7C;FE,1,28,20,25
1760 DATA CD,75,BB,18,EB,CD,78,BB,7C,FE,1,28,12,25,CD,75
1770 DATA BB,CD,60,BB,FE,20,20,ED,CD,78,BB,24,CD,75,BB,C3
1780 DATA 1A,80,CD,60,BB,FE,1,28,F6,CD,78,BB,25,CD,75,BB
1790 DATA 18,BE,CD,70,82,CD,78,BB,25,CD,75,BB,CD,60,BB,FE
1800 DATA 20,20,15,CD,78,BB,7C,FE,4F,28,1E,24,CD,75,BB,CD
1810 DATA 60,BB,FE,20,28,ED,18,11,CD,78,BB,7C,FE,4F,28,9
1820 DATA CD,78,BB,24,CD,75,BB,18,D3,C3,1A,80,CD,70,82,CD
1830 DATA 78,BB,3E,4F,67,CD,75,BB,C3,1A,80,CD,70,82,CD,78
1840 DATA BB,3E,1,67,CD,75,BB,C3,1A,80,0,0,0,0,0,0
1850 '
1860 '-----
1870 ' Modification du 2eme programme
1880 '-----
1890 '
1900 POKE &829E,&63 : POKE &829F,&83
1910 '=====
1920 ' Suite 2 de l'editeur de texte WEKA
1930 '=====
1940 FOR i=1 TO 6 : READ a# : NEXT i 'Lecture des 0

```

```

1950 FOR i=&8363 TO &8576
1960   READ a#
1970   a#="2"+a#
1980   POKE i,VAL(a#)
1990 NEXT i
2000 '
2010 '-----
2020 ' Donnees de la suite 2
2030 '-----
2040 DATA FE,F8,28,12,FE,F9,CA,F2,83,FE,14,CA,5D,84,FE,12
2050 DATA CA,D5,84,C3,2F,82,CD,78,BB,22,87,82,45,5,2A,80
2060 DATA 82,78,B7,28,9,11,50,0,37,3F,ED,52,10,FC,11,80
2070 DATA 7,37,3F,ED,52,22,85,82,11,0,90,ED,52,38,4D,3E
2080 DATA 2,CD,E,BC,2A,85,82,22,80,82,6,19,21,1,1,22
2090 DATA 83,82,C5,E5,6,50,2A,80,82,7E,CD,5A,BB,23,10,F9
2100 DATA 22,80,82,2A,83,82,2C,26,1,22,83,82,CD,75,BB,E1
2110 DATA C1,10,DF,2A,87,82,CD,75,BB,3E,19,95,47,11,50,0
2120 DATA 2A,80,82,37,3F,ED,52,10,FC,22,80,82,C3,1A,80,CD
2130 DATA 78,BB,22,87,82,45,5,2A,80,82,78,B7,28,9,11,50
2140 DATA 0,37,3F,ED,52,10,FC,E5,3E,2,CD,E,BC,E1,11,80
2150 DATA 7,19,22,80,82,6,19,21,1,1,22,83,82,C5,E5,6
2160 DATA 50,2A,80,82,7E,CD,5A,BB,23,10,F9,22,80,82,2A,83
2170 DATA 82,2C,26,1,22,83,82,CD,75,BB,E1,C1,10,DF,2A,87
2180 DATA 82,CD,75,BB,3E,19,95,47,11,50,0,2A,80,82,37,3F
2190 DATA ED,52,10,FC,22,80,82,C3,1A,80,CD,78,BB,22,83,82
2200 DATA 7C,FE,4E,D2,D2,84,C0,70,82,CD,78,BB,25,CD,75,BB
2210 DATA CD,60,BB,FE,20,20,15,CD,78,BB,7C,FE,4F,28,1E,24
2220 DATA CD,75,BB,CD,60,BB,FE,20,28,ED,18,11,CD,78,BB,7C
2230 DATA FE,4F,28,9,CD,78,BB,24,CD,75,BB,18,D3,CD,78,BB
2240 DATA ED,5B,83,82,7C,92,28,27,F5,2A,83,82,CD,75,BB,F1
2250 DATA 47,C5,CD,78,BB,3E,50,94,47,4,16,0,5C,2A,80,82
2260 DATA 19,2B,54,5D,23,ED,A0,10,FC,CD,52,82,C1,10,E2,C3
2270 DATA 1A,80,CD,78,BB,22,83,82,7C,FE,1,CA,74,85,CD,70
2280 DATA 82,CD,78,BB,25,CD,75,BB,CD,60,BB,FE,20,28,1C,CD
2290 DATA 78,BB,7C,FE,1,28,4E,25,CD,75,BB,CD,60,BB,FE,20
2300 DATA 28,2,18,15,CD,60,BB,FE,20,20,E,CD,78,BB,7C,FE
2310 DATA 1,28,32,25,CD,75,BB,18,EB,CD,78,BB,7C,FE,1,28
2320 DATA 24,25,CD,75,BB,CD,60,BB,FE,20,20,ED,CD,78,BB,24
2330 DATA CD,75,BB,18,10,CD,60,BB,FE,1,28,9,CD,78,BB,25
2340 DATA CD,75,BB,18,BF,CD,78,BB,7C,2A,83,82,6F,7C,95,47
2350 DATA B7,28,1E,C5,CD,78,BB,3E,50,94,47,4,16,0,5C,2A
2360 DATA 80,82,19,2B,54,5D,23,ED,A0,10,FC,CD,52,82,C1,10
2370 DATA E2,C3,1A,80,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0
2380 '-----
2390 ' Activation du programme
2400 '-----
2410 CALL &8000

```

Assurez-vous que les données entrées sont correctes grâce au programme de checksum qui doit donner le résultat suivant :

```
62 CE 8C D0 71 DD 24 5C 7B AF AC F1 FE E7 CD 4 2E 27 5A BD
F9 67 C9 D5 7F 9D 48 97 B3 A1 D7 22 A5 1F 33 CB 46 6B 8B E1
DC 34 EE E9 23 24 BB C9 A6 59 78 FD 8D BF 3D BD 9A 8D 56 C4
B1 7A 42 9A F5 61 DF 42 BA FA DA 75 C5 92 C4 1D CF 44 C3 54
C 2A 85 69 4E D0 2C 41
```

Si une des données affichées par le programme de checksum n'est pas identique à celle de même rang dans la liste ci-dessus, la ligne de DATA correspondante contient une ou plusieurs erreurs de frappe...

Pour implémenter les quatre fonctions dont nous venons de parler, procédez comme suit :

- si vous utilisez un assembleur :
 - entrez les codes opératoires donnés dans le premier listing,
 - implantez le programme de base en mémoire,
 - placez les valeurs &H89 et &H82 dans les octets d'adresse &H8051 et &H8052, cela pour effectuer un débranchement dans la boucle principale du programme principal,
 - placez les valeurs &H63 et &H83 dans les octets d'adresse &H829E et &H29F, cela pour effectuer un débranchement dans la boucle principale du premier jeu de fonctions,
 - activez le programme en tapant, sous Basic CALL &8000
- si vous préférez utiliser le chargeur basic :
 - tapez la partie des codes qui correspondent au second jeu de fonctions listées dans le chargeur,
 - exécutez le chargeur.

Résumé :

Les commandes désormais disponibles dans ce traitement de texte sont les suivantes :

- touches flèches pour déplacer le curseur dans toutes les directions ;
- **Carriage Return** pour passer à la ligne ;
- **Ctrl Z** pour effectuer un scrolling d'une ligne vers le haut ;
- **Ctrl W** pour effectuer un scrolling d'une ligne vers le bas ;
- **CLR** pour effacer le caractère courant ;
- **DEL** pour effacer le caractère à gauche du caractère courant ;

- **Ctrl A** pour déplacer le curseur sur la première lettre du mot qui se trouve à droite du mot courant ;
- **Ctrl F** pour déplacer le curseur sur la première lettre du mot qui se trouve à gauche du mot courant ;
- **Ctrl Right** pour déplacer le curseur sur le dernier caractère de la ligne courante ;
- **Ctrl Left** pour déplacer le curseur sur le premier caractère de la ligne courante ;
- **Ctrl Up** pour afficher la page précédente ;
- **Ctrl Down** pour afficher la page suivante ;
- **Ctrl T** pour effacer le mot à droite du curseur ;
- **Ctrl R** pour effacer le mot à gauche du curseur ;
- **Ctrl Q** pour quitter le traitement de texte et retourner sous Basic.